```
(*tropical plot*)
p = 8 x + 2 y + 5 (*a multivariate polynomial*)
```

Out[150]=

$$5 + 8\,x + 2\,y$$

In[154]:=

```
MinPlus[poly_] :=
  ReplaceAll[poly, (*This function takes a polynomial as an argument and applies
    the MinPlus algebra on it*){Plus → Min, Times → Plus, Power → Times}];
MaxPlus[poly_] :=
  ReplaceAll[poly, (*This function takes a polynomial as an argument and applies
    the MinPlus algebra on it*){Plus → Max, Times → Plus, Power → Times}];
```

In[152]:=

```
MinPlus[p] (*applied MinPlus algebra on a polynomial p*)
```

Out[152]=

$$\text{Min}[5,\ 8 + x,\ 2 + y]$$

In[153]:=

```
ContourPlot[MinPlus[p], {x, -5, 2}, {y, -2, 5}, ExclusionsStyle → Black]
```

Out[153]=



In[156]:=

```
MaxPlus[p]
```

Out[156]=

$$\text{Max}[5,\ 8 + x,\ 2 + y]$$

In[158]:=

```
ContourPlot[MaxPlus[p], {x, -6, 0}, {y, 0, 6}, ExclusionsStyle → Black]
```

Out[158]=



In[159]:=

```
ContourPlot[MaxPlus[p], {x, -6, 0}, {y, 0, 6},
 ContourShading → None, ExclusionsStyle → Black]
```

Out[159]=



In[160]:=

```
ContourPlot[MaxPlus[p], {x, -6, 0}, {y, 0, 6},
 ContourStyle → Directive[GrayLevel[0], Opacity[0.], AbsoluteThickness[0.], Dotted],
 ContourShading → None, ExclusionsStyle → Black]
```

Out[160]=



In[161]:=

```
(*tropicalization*)
valuation = IntegerExponent[#, 2] & /@ Last /@ CoefficientRules[p, {x, y}]
   (*this function brings the valuation of every coefficient*)
```

Out[161]=

```
{3, 1, 0}
```

In[166]:=

```
TropicalMinPlusPoly[poly_, variables_] :=
 Min @@ (Flatten /@ Transpose[{First /@ CoefficientRules[poly, variables],
       IntegerExponent[#, 2] & /@ Last /@
         CoefficientRules[poly, variables]}].Append[variables, 1])
TropicalMaxPlusPoly[poly_, variables_] :=
 Max @@ (Flatten /@ Transpose[{First /@ CoefficientRules[poly, variables],
       IntegerExponent[#, 2] & /@ Last /@
         CoefficientRules[poly, variables]}].Append[variables, 1])
```

In[165]:=

```
TropicalMinPlusPoly[p, {x, y}]
```

Out[165]=

```
Min[0, 3 + x, 1 + y]
```

In[170]:=
```
f = TropicalMaxPlusPoly[p, {x, y}]
```
Out[170]=
```
Max[0, 3 + x, 1 + y]
```
In[172]:=
```
ContourPlot[f, {x, -10, 10}, {y, -10, 10}, ContourStyle →
  Directive[GrayLevel[0], Opacity[0.`], AbsoluteThickness[0.`], Dotted],
 ContourShading → None, ExclusionsStyle → Black]
```
Out[172]=



In[173]:=
```
(*Newton Polytope*)
NewtonPolytope[poly_, variables_] :=
 Module[{tpoly, p, cs, P2 = poly}, p = First /@ CoefficientRules[P2, variables];
  cs = IntegerExponent[#, 2] & /@ Last /@ CoefficientRules[P2, variables];
  tpoly = Flatten /@ Transpose[{p, cs}];
  Show[ListPlot3D[tpoly, Mesh → All, Axes → True,
    PlotStyle → {Yellow}, Filling → Bottom], ListPointPlot3D[tpoly]]]
```

In[174]:=

```
NewtonPolytope[16 x^2 + 6 x y + 7 y^2 + 7 x + 5 y + 2, {x, y}]
```

Out[174]=



In[175]:=

```
NewtonPolytope[p, {x, y}]
```

Out[175]=



In[224]:=
```
(*Puiseux series and valuations*)
PuiseuxValuation[expr_] :=
  Exponent[expr, t, List] /. {e : {__} :> Min[e], {} -> -Infinity};

PuiseuxValuation[1 / t - 4 / t^2 + 5 t^2 + t^20]
PuiseuxValuation[8 t^6 + t^20 + t^7]
PuiseuxValuation[0]
PuiseuxValuation[30 t^(94 / 3) - 59 / (t^(98 / 5))]
(*-2 6-∞ -98/5*)
```

Out[225]=
$$-2$$

Out[226]=
$$6$$

Out[227]=
$$-\infty$$

Out[228]=
$$-\frac{98}{5}$$

In[263]:=
```
TropicalPuiseuxPoly[poly_, variables_] :=
 Max @@ (Flatten /@ Transpose[{First /@ CoefficientRules[poly, variables],
       -PuiseuxValuation /@ MonomialList[poly, variables]}].Append[variables, 1])
```

In[198]:=
```
f = (t^(-1) + 1) x + (t^(2) - 3 t^(3)) y + 5 t^(4)
PuiseuxValuation /@ MonomialList[f, {x, y}]
```

Out[198]=
$$5\,t^4 + \left(1 + \frac{1}{t}\right) x + \left(t^2 - 3\,t^3\right) y$$

Out[199]=
$$\{-1, 2, 4\}$$

In[264]:=
```
ff = TropicalPuiseuxPoly[f, {x, y}]
```

Out[264]=
$$\text{Max}[-4, 1 + x, -2 + y]$$

In[265]:=

```
(*Exercises*)
ContourPlot[ff, {x, -10, 10}, {y, -10, 10}, ContourStyle →
  Directive[GrayLevel[0], Opacity[0.`], AbsoluteThickness[0.`], Dotted],
 ContourShading → None, ExclusionsStyle → Black]
```

Out[265]=



In[259]:=

```
g = t^(3) x^(2) + 2 x * y + t * y^(2) + (t + t^(3)) * x - 5 y + 1
```

Out[259]=

$$1 + \left(t + t^3\right)\, x + t^3\, x^2 - 5\, y + 2\, x\, y + t\, y^2$$

In[266]:=

```
gg = TropicalPuiseuxPoly[g, {x, y}]
```

Out[266]=

$$\text{Max}[0,\ -1 + x,\ -3 + 2\, x,\ y,\ x + y,\ -1 + 2\, y]$$

In[272]:=

```
ContourPlot[gg, {x, -2, 3}, {y, -2, 2}, ContourStyle →
  Directive[GrayLevel[0], Opacity[0.`], AbsoluteThickness[0.`], Dotted],
 ContourShading → None, ExclusionsStyle → Black]
```

Out[272]=